LabSmith Application Note

Continuous Flow with uProcess™ Automation

By Nicholas Carrano and Michael Duperly
LabSmith, Inc., Livermore, CA 94550

Biological experiments often require continuous fluid delivery over experiment times of hours to days. Many experiments also require small platforms, and small syringe pumps with high flow rate accuracy. LabSmith's uProcess™ automated fluid routing system meets these challenges. This application note describes the components and layout for a uProcess continuous flow application using a four port, two-position AV202 valve and offers script options for optimal performance.

Introduction

An automated uProcess™ program can control the withdrawal of reagent to fill one syringe pump while simultaneously dispensing the fluid to an experiment from a second syringe pump. The AV202 valve switches between the full and empty syringes for uninterrupted fluid delivery at precise flow rates.

Table 1 shows the required equipment. These components are included in the **Continuous Dispense Syringe Kit**.

Table 1. Components for Continuous Dispense Kit

Part Number	Description	Quantity
EIB200	Electronic interface controller with uProcess™ software	**1**
AV202	Two-position, 4-Port Automated Valve	**1**
4VM02	Manifold for control of up to 4 automatic valves	**1**
SPS01	SPS01 syringe pump with syringe glass and plunger set	**2**
uPB-05	uProcess™ breadboard with 5 device connections	**1**
BBRES-5ML	Breadboard reservoir	**1**
C360-100, T132-100, or T116-100	One-piece fitting for connecting capillary	**10**
C360-101 or T116-101	One-piece plug	**2**
CAP360, TUBE132, or TUBE116	Tubing/capillary	**1**
SPS-TOOLS, LS-HEX, LS-TORX, LS-Screws.25	Component & breadboard installation tools	**1 Ea.**

Figure 1 shows the breadboard layout.



Figure 1. uProcess hardware: AV202 series automated valve, SPS01 programmable syringe pumps, 4VM02 valve, 4VM02 manifold, and uPB-05 breadboard

uProcess automation and software can provide uninterrupted delivery of fluid at precise flow rates, using two SPS01 syringe pumps and a four port, two position valve (AV202). Figure 2 shows the fluid routing configurations of the AV202 valve.



Figure 2. AV202 Fluid routing configurations

uProcess Automation

The example below shows a simple uProcess continuous dispense script. The first syringe dispenses fluid while the other syringe is filling. Once the first syringe finishes dispensing, the valve switches to dispense fluid from the second syringe while refilling the first syringe.

Example uProcess Continuous Dispense Script V1

```
*Syringe1 =       SPS 80 ul
*Syringe2 =       SPS 80 ul
*4VM =            4VM

Step1:            ; Syringe1 dispensing,
Syringe2 filling
         4VM:    SetValves(3, 0, 0, 0)
         Syringe1:        SetFlowRate(100.0
ul/min)
         Syringe1:        MoveTo(1.0 ul)
         Syringe2:        SetFlowRate(300.0
ul/min)
         Syringe2:        MoveTo(80.0 ul)
         WaitDone()

Step2:            ; Syringe2 dispensing,
Syringe1 filling
         4VM:    SetValves(1, 0, 0, 0)
         Syringe1:        SetFlowRate(300.0
ul/min)
         Syringe1:        MoveTo(80.0 ul)
         Syringe2:        SetFlowRate(100.0
ul/min)
         Syringe2:        MoveTo(1.0 ul)
         WaitDone()
         Loop Step1 25
```

For this example, a flow dampener (**BBRES-1ML**) was placed downstream of the valve to remove the pulsations caused by the SPS01 syringe pumps. A Sensirion® Flow Sensor was used to measure the flow rate of fluid exiting the flow dampener. Figure 3 shows that, while this approach maintains a steady flow for the majority of the run time, each time the valve switches there is a sudden drop in the flow rate while the second syringe starts up.



Figure 3. Flow rate plot V1

The next section describes techniques to reduce the flow disturbance when the system switches syringe pumps.

Tips to Optimize Flow Profiles

One cause of the flow disturbance during switching is the syringe start-up time. The syringe motor includes an anti-backlash spring, and when the syringe direction changes from pull to push, the spring can cause a slight delay before the syringe plunger starts advancing. The duration of the delay will vary depending on the syringe size, set speed, system back pressure, and fluid viscosity. To eliminate this delay, modify the script to dispense solution back into the reservoir before the valve switch occurs, as shown in Script V2.

Example uProcess Continuous Dispense Script V2

```
*Syringe2 =       SPS 80 ul
*Syringe1 =       SPS 80 ul
*4VM =            4VM

Step1:
         4VM:    SetValves(3, 0, 0, 0)
         Syringe1:        SetFlowRate(100.0
ul/min)
         Syringe1:        MoveTo(1.0 ul)
         Syringe2:        SetFlowRate(300.0
ul/min)
         Syringe2:        MoveTo(80.0 ul)
         Syringe2:        WaitDone()
         Syringe2:        MoveTo(75.0 ul)
         Syringe1:        WaitDone()

Step2:
         4VM:    SetValves(1, 0, 0, 0)
         Syringe2:        SetFlowRate( 100.0
ul/min)
         Syringe2:        MoveTo(1.0 ul)
         Syringe1:        SetFlowRate(300.0
ul/min)
         Syringe1:        MoveTo(80.0 ul)
         Syringe1:        WaitDone()
         Syringe1:        MoveTo(75.0 ul)
         Syringe2:        WaitDone()
         Loop Step1 25
```

Figure 4 shows a plot of the measured flow rate over time. Note the reduction in sudden flow rate drops.



Figure 4. Flow rate plot V2

While the previous example reduced the flow disturbance when the valve switched, the ramp up of the syringe flow rate is still evident when the valve switches. To resolve this issue, start the flow of the second syringe before the first syringe stops dispensing and the valve switches, as shown in the example below.

Example uProcess Continuous Dispense Script V3

```
*Syringe1 =  SPS 80 ul
*Syringe2 =  SPS 80 ul
*4VM =       4VM

Setup:
    4VM: SetValves(3, 0, 0, 0)
    Syringe2:  SetFlowRate(300.0 ul/min)
    Syringe2:  MoveTo(80.0 ul)
    Syringe2:  WaitDone()
    4VM:  SetValves(1, 0, 0, 0)
    Syringe2:  SetFlowRate(80.0 ul/min)
    Syringe2:  MoveTo(1.0 ul)

Step1:
    Syringe1: SetFlowRate(300.0 ul/min)
    Syringe1: MoveTo(80.0 ul)
    Syringe1: WaitDone()

Step2:
    if (Syringe2.Volume <= 10 ul)
    {
       Syringe1:     SetFlowRate(80.0 ul/min)
       Syringe1:     MoveTo(1.0 ul)
       Wait(2 s)
       4VM:     SetValves(3,0,0,0)
       Wait(3 s)
       Goto Step3
    }
    Wait(500 ms)
    Goto Step2

Step3:
    Syringe2:    SetFlowRate(300.0 ul/min)
    Syringe2:    MoveTo(80.0 ul)
    Syringe2:    WaitDone()

Step4:
    if (Syringe1.Volume <= 10 ul)
    {
        Syringe2:    SetFlowRate(80.0 ul/min)
        Syringe2:    MoveTo(1.0 ul)
        Wait(2 s)
        4VM:    SetValves(1,0,0,0)
        Wait(3 s)
        Goto Step1
    }
    Wait(500 ms)
    Goto Step4
```

Figure 5 shows a plot of the measured flow rate over time. Note that the transitions between syringes are nearly unidentifiable.



Figure 5. Flow rate plot V3

Conclusion

Comparing the methods shown above, if smooth flow transitions are important for the application, begin dispensing the filled syringe before switching the valve, and switch the valve as the first syringe is still dispensing. The script is slightly more complicated and will result in more frequent switching of the valve, but this technique can provide essentially smooth transitions between flows.



Figure 6. Superimposed flow rate plots

For more information on creating smooth flow with the SPS01 syringe pumps see the LabSmith Application Note, **Pressure Driven Flow with uProcess™ Devices and Software**.

For questions, please call +1 (925) 292-5161 or email sales@labsmith.com. LabSmith is a trademark of LabSmith, Inc. Sensirion is a trademark of Sensirion AG. All specifications are subject to change without notice.